

H26年度 HPFプログラミング入門

2014年 6月26日

大阪大学サイバーメディアセンター
日本電気株式会社

**本資料は、大阪大学サイバーメディアセンターとNECの共同により作成されたものです。
無断転載等は、ご遠慮下さい。**

目次

1. HPF概要
2. HPFの基本機能
3. HPFの基本機能まとめ
4. HPFプログラミング演習(1)
5. 並列化情報リストの利用
6. クリーンナップとデバッグ
7. 実行時性能情報の取得方法
8. HPFプログラミング演習(2)
9. HPFプログラミングのテキスト紹介


1. HPF概要

並列処理・並列実行

- 仕事 (処理) を複数のコアまたはCPUに分割し、同時に実行すること

並列化

- 並列処理を可能とするために、処理の分割を行うこと



並列化の効果

行列積プログラム

```
implicit real(8)(a-h,o-z)
parameter ( n=15360 )
real(8) a(n,n),b(n,n),c(n,n)
real(4) etime,cp1(2),cp2(2),t1,t2,t3
do j = 1,n
  do i = 1,n
    a(i,j) = 0.0d0
    b(i,j) = n+1-max(i,j)
    c(i,j) = n+1-max(i,j)
  enddo
enddo
write(6,50) ' Matrix Size = ',n
50 format(1x,a,i5)
t1=etime(cp1)
do j=1,n
  do k=1,n
    do i=1,n
      a(i,j)=a(i,j)+b(i,k)*c(k,j)
    end do
  end do
end do
t2=etime(cp2)
t3=cp2(1)-cp1(1)
write(6,60) ' Execution Time = ',t2,' sec',' A(n,n) = ',a(n,n)
60 format(1x,a,f10.3,a,1x,a,d24.15)
stop
end
```


● SX-9 1CPUの実行時間は約74.4秒

```
% ./a.out
Matrix Size = 15360
Execution Time =      74.376 sec  A(n,n) =      0.153600000000000D+05

***** Program Information *****
Real Time (sec)           :      74.317164
User Time (sec)           :      74.307155
Sys Time (sec)            :       0.008041
Vector Time (sec)         :      74.306276
Inst. Count               :    56232203727.
V. Inst. Count            :    35849130393.
V. Element Count          :    9175961798122.
FLOP Count                :    7247757312124.
MOPS                      :    123761.229608
MFLOPS                    :    97537.811966
A. V. Length              :    255.960513
V. Op. Ratio (%)          :    99.778357
Memory Size (MB)          :    5568.031250
MIPS                      :    756.753555
I-Cache (sec)             :       0.000406
O-Cache (sec)             :       0.000538
Bank Conflict Time
  CPU Port Conf. (sec)    :       0.012855
  Memory Network Conf. (sec) :       2.405
```

並列化の効果

- 並列化により複数のCPUを利用し、実行時間を短縮
- 分散並列を用いることで、SX-9の複数ノードが利用可能



1ノードでは4.9秒まで。さらに時間を短縮するためには複数ノード必要。

SX-9のプログラミング環境

1ノード

共有メモリ
並列処理

(各CPU)

ベクトル処理

自動並列

自動並列化

手動並列

OpenMP

自動

自動ベクトル化

マルチノード

共有メモリ
並列処理

(1ノード内でも利用可)

データ分割
指示型(自動)

HPF

メッセージ
交換型(手動)


MPI

HPFとは

■ HPF(High Performance Fortran)とは？

- Fortranの分散メモリ型並列計算機向け**拡張仕様**で以下の特徴を有する
- 国際的な標準仕様(並列処理の主要ベンダ, 大学, 研究機関が共同で仕様策定)
- 記述が平易(Fortran＋コメント形式の指示文)
- 上級ユーザは細かな制御も可能(通信の制御, 部分的にMPIを利用したチューニング等)

分散並列化のために必要な作業



HPFとMPI

■ HPFは、分散並列プログラム開発の3つの局面

- ① 複数プロセッサへの**データの分割配置(マッピング)**
- ② 複数プロセッサへの**並列化・処理の分担**
- ③ 複数プロセッサ間の**通信処理の挿入**

のうち、②と③を**自動化**

	MPI	HPF
①データの分割配置	手動	手動
②並列化・処理の分担	手動	自動
③通信処理の挿入	手動	自動

※MPI(Message Passing Interface)

HPFとMPI

1,000行程度の流体解析コードの並列化による比較

		SX-9/1CPU	SX-9/16CPU
オリジナルコード	約1,000行	1,706秒	
MPI化コード	すべて手動で並列化. 約580行の追加・修正必要		124秒
HPF化コード	約40行の追加のみ		137秒

■ 上記コードの場合, 少ないプログラム修正(追加)でMPI化コードと比較して遜色のない並列効果を達成することが可能.

コンパイル・実行コマンド

HPFプログラムのコンパイル

`sxhpf [オプション] ソースファイル名`

※オプションはsxf90と同様. SX-9向けオブジェクト作成では「-cfsx9」が必要.

HPFプログラムの実行(基本的にはMPIプログラムと同じ)

`mpirun [-host ホスト番号] -np プロセス数ロードモジュール名`

※「-host」はマルチノードで使用する場合に必要(シングルノードでは必要ない)

※ホスト番号は「0」から始めて、「1」、「2」...と使用する「ノード数-1」まで指定.

「-host」と「-np」の組み合わせはノード数分必要.

※プロセス数は使用するジョブクラスに応じて、以下のように指定

ジョブクラス	プロセス数	メモリ容量	マシン
DBG	4	16GB	SX-8R
SX8F	8	120GB	
SX8L	32	1TB	
DGB9	4	128GB	SX-9
SX9	64	4TB	

※左記のプロセス数はHPFのみで並列化を行う場合のもの. 共有メモリ並列との併用の場合(-Pautoオプションの追加)は, 左記のプロセス数／共有メモリ並列数の値を-npで指定(F_RSVMTASKで指定する共有メモリ並列数×ノード当たりのプロセス数がCPU数を超えないこと).

※左記のメモリ容量は使用するジョブクラスの最大利用可能なサイズ


2. HPFの基本機能

プロセッサの構成を宣言する	PROCESSORS指示文
配列を分散する	DISTRIBUTE指示文
並列化可能であることを明示する	INDEPENDENT指示文 [オプション] ◆作業変数指定: NEW節 ◆集計変数指定: REDUCTION節
実行プロセッサを指定する	ON指示構文 [オプション] ◆通信不要を明示: LOCAL節


PROCESSORS指示文

プロセッサの構成を宣言する

```
!HPF$ PROCESSORS P(N)
```



```
!HPF$ PROCESSORS P(N,M)
```



書式: プロセッサ構成 p_i を宣言する(配列の宣言と同様)

PROCESSORS p_1 (上下限,...), p_2 (上下限,...), ...

特徴:

プロセッサ構成の各要素は**抽象プロセッサ**と呼ばれ、HPF/SX V2では**プロセスと一対一**に対応する

組込み関数 `number_of_processors()` を使用して、プログラムを修正せずに、プロセッサ数を実行時に決定することも可能

例:

```
!HPF$ PROCESSORS P(number_of_processors())
```

省略すると、実行時のプロセス数から、**処理系が自動的に決定する**

➤ 多くの場合、PROCESSORS指示文の指定を省略しても特に問題ない

2. HPFの基本機能

プロセッサの構成を宣言する	PROCESSORS指示文
配列を分散する	DISTRIBUTE指示文
並列化可能であることを明示する	INDEPENDENT指示文 [オプション] ◆作業変数指定: NEW節 ◆集計変数指定: REDUCTION節
実行プロセッサを指定する	ON指示構文 [オプション] ◆通信不要を明示: LOCAL節

DISTRIBUTE指示文(1)

■ 配列のプロセッサ構成へのマップ方法(**分散**)を指定する



書式: 配列 t_i をプロセッサ構成 p 上に分散する

◆ 1つの配列の分散([]内は省略可能)

DISTRIBUTE t_1 (分散方法,...) [ONTO p]

◆ 複数の配列の分散([]内は省略可能)

DISTRIBUTE (分散方法,...) [ONTO p] :: $t_1, t_2, \dots$

分散方法

● 5種類の分散方法が指定可能

BLOCK、*、GEN_BLOCK、CYCLIC、INDIRECT

DISTRIBUTE指示文(2) BLOCK分散


連続した配列要素を均等に分散する

- 分散ブロック幅 = $(\text{配列要素数} - 1) / \text{プロセッサ数} + 1$
- BLOCK(m)の指定によりm要素ずつの分散も可能

用途 負荷が要素に対して均一で、近接参照が多い場合に適する

例:

```
!HPF$ PROCESSORS P(4)  
  REAL A(19)  
!HPF$ DISTRIBUTE A(BLOCK) ONTO P
```




DISTRIBUTE指示文(3) GEN_BLOCK分散

■ 連続した配列要素を不均等に分散する

用途 要素に対する負荷バランスが不均等な場合に適する

例:

```
!HPF$ PROCESSORS P(4)  
  REAL A(19)  
  INTEGER, PARAMETER:: M(4)=(/6,3,5,5/)  
!HPF$ DISTRIBUTE A(GEN_BLOCK(M)) ONTO P
```



手続境界での引数のマッピング

■ 実引数と仮引数のマッピングは独立に指定可能

例:

```
!HPF$ PROCESSORS P(4)
!HPF$ DISTRIBUTE A(BLOCK,*) ONTO P
      INTEGER A(N,N)
      CALL SUB(A,N)
      .
      END

      SUBROUTINE SUB(A,N)
!HPF$ PROCESSORS P(4)
!HPF$ DISTRIBUTE A(*,BLOCK) ONTO P
      INTEGER A(N,N)
```

実引数A

P(1)
P(2)
P(3)
P(4)

SUBの呼出し時 ↓ ↑ SUBからの戻り時

仮引数A

P(1)	P(2)	P(3)	P(4)

実引数と仮引数のマッピングが異なる場合

- ◆ 手続の呼出し時・戻り時にそれぞれ自動的に通信が発生する(合計2回)
 - 呼出し時は、実引数のマッピングを仮引数と合わせるため
 - 戻り時は、仮引数のマッピングを実引数のマッピングに戻すため

2. HPFの基本機能

プロセッサの構成を宣言する	PROCESSORS指示文
配列を分散する	DISTRIBUTE指示文
並列化可能であることを明示する	INDEPENDENT指示文 [オプション] ◆作業変数指定: NEW節 ◆集計変数指定: REDUCTION節
実行プロセッサを指定する	ON指示構文 [オプション] ◆通信不要を明示: LOCAL節

INDEPENDENT指示文(1)

■ 直後のループが並列実行できることを、コンパイラに教える

```
!HPF$ INDEPENDENT
```

```
  DO I = 1, N  
    A(I)=B(I)+C(I)  
  END DO
```

書式: ([]内は省略可能)

INDEPENDENT [, NEW(変数名,...)] [, REDUCTION(変数名,...)]

オプションとして、以下の2つが指定可能

NEW節	作業変数
REDUCTION節	集計変数

用途

並列化できるにもかかわらず、コンパイラが並列化できないと判定したループだけに指定すればよい (多くの場合、INDEPENDENT指示文なしでも、コンパイラが自動的に並列化可能性を判定できる)

INDEPENDENT指示文(2)

並列化できないループの例 (実行順序により結果が変わる場合)

- ある繰返しで**定義**された要素を他の繰返しで定義・使用すると並列化を阻害する(データ依存)
- このようなループにはINDEPENDENT指示文は指定できない(指定すると結果不正となる)

DO I = 2,N A(I)=A(I)+A(I-1) END DO	$\begin{array}{l} A(2) = A(2) + A(1) \\ \quad \searrow \\ A(3) = A(3) + A(2) \end{array}$	依存: 定義後に使用
DO I = 1,N-1 A(I)=A(I)+A(I+1) END DO	$\begin{array}{l} A(1) = A(1) + A(2) \\ \quad \swarrow \\ A(2) = A(2) + A(3) \end{array}$	逆依存: 使用後に定義
DO I = 1,N IF (A(I)>0) S=A(I) END DO	$\begin{array}{l} S = A(1) \\ \downarrow \\ S = A(2) \end{array}$	出力依存: 定義後に定義
DO I = 1,N IF (A(I)>0) STOP END DO	・ループ中のSTOP文 ・ループからの飛び出し	制御依存: 途中で実行が終わってしまう可能性がある


INDEPENDENT指示文(3) REDUCTION節(1)

集計計算	総和・最大・最小のように、どの順序で実行しても結果が(数学的には)同じになる計算
REDUCTION節	集計計算を行うループにINDEPENDENT指示文を指定する場合に、併せて指定する (自動で並列化された場合は不要)

例:

```
!HPF$ DISTRIBUTE A(BLOCK)
      INTEGER A(N),S
      DO I=1,N
        S = S + A(I)
      ENDDO
```

並列化対象



```
!HPF$ DISTRIBUTE A(BLOCK)
      INTEGER A(N),S
!HPF$ INDEPENDENT, REDUCTION(S)
      DO I=1,N
        S = S + A(I)
      ENDDO
```

並列化対象

制限: 集計計算(総和・最大・最小・...)は、以下の形式で記述しなければならない

$A = A + E$	$A = E .and. A$	$A = iand(A, E1, ..., En)$	A: 集計変数 E, Ei: 式
$A = E + A$	$A = A .or. E$	$A = ior(A, E1, ..., En)$	
$A = A - E$	$A = E .or. A$	$A = ieor(A, E1, ..., En)$	
$A = A * E$	$A = A .eqv. E$	$A = min(A, E1, ..., En)$	
$A = E * A$	$A = E .eqv. A$	$A = max(A, E1, ..., En)$	
$A = A / E$	$A = A .neqv. E$		
$A = A .and. E$	$A = E .neqv. A$		

INDEPENDENT指示文(4) REDUCTION節(2)

■集計種別付き REDUCTION節(HPF/JA拡張)

用途

任意の形式の集計計算を含むループに、INDEPENDENT指示文を指定する場合に指定する(**自動で並列化された場合は不要**)

書式: ([]内は省略可能)

REDUCTION(**集計種別**: 集計変数名 [**/ 位置変数名 ,.../**],...)

集計種別

+, *, max, min, firstmax, firstmin, lastmax, lastmin, iand, ior, ieor, .and., .or., .eqv., .neqv.

例:

!HPF\$ DISTRIBUTE A(BLOCK)
INTEGER A(N)
!最大値と最初に見つかった場所
DO i=1,N
if(m.lt.a(i))then
m = a(i)
idx = i
endif
ENDDO

並列化対象

!HPF\$ DISTRIBUTE A(BLOCK)
INTEGER A(N)
!hpf\$ independent,reduction(firstmax: m /idx/)
DO i=1,N
if(m.lt.a(i))then
m = a(i)
idx = i
endif
ENDDO

並列化対象

2. HPFの基本機能

プロセッサの構成を宣言する	PROCESSORS指示文
配列を分散する	DISTRIBUTE指示文
並列化可能であることを明示する	INDEPENDENT指示文 [オプション] ◆作業変数指定: NEW節 ◆集計変数指定: REDUCTION節
実行プロセッサを指定する	ON指示構文 [オプション] ◆通信不要を明示: LOCAL節

ON指示構文(1) 実行プロセッサの指定

書式: 部分配列 がマップされているプロセッサが実行する

◆1つの文・構文に対して([]内は省略可能)

ON HOME(部分配列) [, LOCAL[(配列名)]]

◆複数の文・構文に対して([]内は省略可能)

ON HOME(部分配列) [, LOCAL[(配列名)]] BEGIN

: 実行文・実行構文の列

ENDON

実際上は、LOCALとセットで利用する場合がほとんど

例:

```
!HPF$ PROCESSORS P(4)
  REAL A(19)
!HPF$ DISTRIBUTE A(BLOCK) ONTO P
```

```
!HPF$ ON HOME(A(1))
```


```
  A(1) = 0
```

```
!HPF$ ON HOME(A(19)) BEGIN
```

```
  A(18) = 1
```

```
  A(19) = 0
```

```
!HPF$ ENDON
```



ON指示構文(2) 通信不要の明示


LOCAL節	- 指定した変数に対する通信が不要であることをコンパイラに教える - 変数指定を省略すると、全ての変数が通信不要であることを意味する
主な用途	無駄な通信が発生している(コンパイラが通信が不要であることを見切れてない)場合に指定する

例:

```
!HPF$ PROCESSORS P(4)  
  REAL A(19)  
!HPF$ DISTRIBUTE A(BLOCK) ONTO P
```


```
!HPF$ ON HOME(A(1)), LOCAL(A)  
  A(1) = 0
```

```
!HPF$ ON HOME(A(19)), LOCAL BEGIN  
  A(18) = 1  
  A(19) = 0  
!HPF$ ENDON
```



ON指示構文(3) 並列ループへの指定例

```
REAL A(10)
!HPF$ PROCESSORS P(2)
!HPF$ DISTRIBUTE A(BLOCK) ONTO P
DO I=1,10
  A(I) = I
END DO
```



```
REAL A(10)
!HPF$ PROCESSORS P(2)
!HPF$ DISTRIBUTE A(BLOCK) ONTO P
DO I=1,10
  !HPF$  ON HOME(A(I)), LOCAL BEGIN
    A(I) = I
  !HPF$  END ON
END DO
```

P(1)					
I =	1	2	3	4	5
A	1	2	3	4	5

P(2)					
I =	6	7	8	9	10
A	6	7	8	9	10

- DO Iの繰返しは、**A(I)を保持するプロセッサが実行すると、通信が不要であることを指定する**
- 通常は、コンパイラが自動的に通信が最小になるよう並列化するため不要。**無駄な通信が発生している場合に指定する。**



ON指示構文(4) 境界処理ループへの指定例

```
real a(10,10)
!hpf$ PROCESSORS p(2)
!hpf$ DISTRIBUTE (*,BLOCK) ONTO p :: a
do i=1,10
  a(i,1)=a(i,2)
enddo
```



```
real a(10,10)
!hpf$ PROCESSORS p(2)
!hpf$ DISTRIBUTE (*,BLOCK) ONTO p :: a
!hpf$ ON HOME(a(:,1)), LOCAL BEGIN
do i=1,10
  a(i,1)=a(i,2)
enddo
!hpf$ ENDON
```


分散されていない次元は"."を指定する



- 例えば、2プロセッサで実行する場合、a(:,1)を保持するプロセッサ(p(1))だけで実行すると、通信なしで実行可能。これをLOCAL節で明示する。(一方、10プロセッサで実行する場合は通信が必要なので、LOCAL節は指定できない)
- ON指示構文とLOCAL節の指定がないと、p(2)も実行に参加し、通信が発生する

3. HPFの基本機能まとめ

プロセッサ構成を宣言する	PROCESSORS指示文
配列を分散する	DISTRIBUTE指示文
並列化可能であることを明示する	INDEPENDENT指示文 [オプション] ◆作業変数指定: NEW節 ◆集計変数指定: REDUCTION節
実行プロセッサを指定する	ON指示構文 [オプション] ◆通信不要を明示: LOCAL節



規則的な問題なら、大半のプログラムがこれだけで並列化可能

3. HPFの基本機能まとめ

HPFプログラミングの基本手順

- 1 コードのクリーンナップ
 - ・ベクトル化のチューニングは、並列化前に済ませておく
- 2 どのループを**並列化する**か決める
 - ・一番実行コストの高いループ
- 3 配列の**マッピング**を決める
 - ・並列化するループに対応する次元でマップする
- 4 HPFコンパイラで**翻訳**してみる
- 5 **並列化情報リスト・診断メッセージ**から指示文を追加・修正する
 - ・配列のマッピングを追加・修正する
 - ・並列化可能であることを明示する指示文(INDEPENDENT)を追加する
 - ・通信が不要であることを明示する指示文(ON+LOCAL)を追加する
 - ・その他の指示文や、F90/SX指示行を追加する
- 6 実行性能を確認し、上記の手順4・5を繰り返す

4. HPFプログラミング演習(1)

■ 姫野ベンチマークコード(himenoBMTsp_s.f)を用いて, HPFプログラミングの流れを学習する.

※姫野ベンチマーク:ポアッソン方程式解法をヤコビの反復法で解く場合の主要なループの処理速度を計るものである(<http://accc.riken.jp/2145.htm>).

※本講習では上記Webページからダウンロードしたソースコードの一部修正したものを用いる.

4. HPFプログラミング演習(1) Step-1

■ プログラムをコンパイルして実行する

- 姫野ベンチマークコードをhimeno.fとして用意.
- himeno.fはFortranで記述した逐次実行プログラムなのでFORTRAN90/SXコンパイラでコンパイルする.
- ディレクトリを移動する.

```
% cd hpf/practice_1
```

- FTRACE情報を採取して演算コストを把握する.

```
% sxf90 -ftrace -cfsx9 himeno.f
```

- 実行スクリプトを用意していますので, サブミットしてSX-9で実行する.

```
% qsub run.sh
```

4. HPFプログラミング演習(1) Step-1

実行結果を確認

- 実行結果はpractice_1.oXXXX(XXXXはジョブID)として格納される.

% cat practice_1.oXXXX

```
mimax= 513  mjmax= 257  mkmax= 257
imax= 512  jmax= 256  kmax= 256
Start rehearsal measurement process.
Measure the performance in 1000 times.
MFLOPS: 25922.51    time(s): 43.15582895278931  0.2500000
```

- FTRACE情報を確認する.

% sxftace

PROC.NAME	FREQUENCY	EXCLUSIVE TIME[sec](	%)	AVER.TIME [msec]	MOPS	MFLOPS	V.OP RATIO	AVER. V.LEN	VECTOR TIME	I-CACHE MISS	O-CACHE MISS	BANK CPU	CONFLICT PORT	NETWORK
jacobi	1	43.153(	99.9)	43152.759	57447.7	26308.6	99.57	251.7	43.153	0.000	0.000	0.289	13.163	
initmt	1	0.038(	0.1)	37.796	27808.4	1.7	99.40	256.0	0.038	0.000	0.000	0.002	0.000	
main_	1	0.000(	0.0)	0.324	220.2	0.4	48.24	228.2	0.000	0.000	0.000	0.000	0.000	

total	3	43.191(100.0)		14396.960	57421.4	26285.4	99.57	251.7	43.190	0.000	0.000	0.291	13.163	

4. HPFプログラミング演習(1) Step-2

■ 演算コストが集中しているのはサブルーチンjacobi

- サブルーチンjacobiのどのループを並列化するか検討する.

```
176: +----->      DO loop=1,nn
177: +----->          DO K=2,kmax-1
178: +----->              DO J=2,jmax-1
179: V----->                  DO I=2,imax-1
180:      A              S0=a(I,J,K,1)*p(I+1,J,K)+a(I,J,K,2)*p(I,J+1,K)
181:      1              +a(I,J,K,3)*p(I,J,K+1)
182:      2              +b(I,J,K,1)*(p(I+1,J+1,K)-p(I+1,J-1,K)
183:      3              -p(I-1,J+1,K)+p(I-1,J-1,K))
184:      4              +b(I,J,K,2)*(p(I,J+1,K+1)-p(I,J-1,K+1)
185:      5              -p(I,J+1,K-1)+p(I,J-1,K-1))
186:      6              +b(I,J,K,3)*(p(I+1,J,K+1)-p(I-1,J,K+1)
187:      7              -p(I+1,J,K-1)+p(I-1,J,K-1))
188:      8              +c(I,J,K,1)*p(I-1,J,K)+c(I,J,K,2)*p(I,J-1,K)
189:      9              +c(I,J,K,3)*p(I,J,K-1)+wrk1(I,J,K)
190:      A              SS=(S0*a(I,J,K,4)-p(I,J,K))*bnd(I,J,K)
191:      GOSA=GOSA+SS*SS
192:      A              wrk2(I,J,K)=p(I,J,K)+OMEGA *SS
193:      V----->          enddo
194:      +----->      enddo
195:      +----->  enddo
197: +----->      DO K=2,kmax-1
198: +----->          DO J=2,jmax-1
199: V----->              DO I=2,imax-1
200:      A              p(I,J,K)=wrk2(I,J,K)
201:      V----->          enddo
202:      +----->      enddo
203:      +----->  enddo
205: +----->  enddo
```

4. HPFプログラミング演習(1) Step-2

■ データマッピングの検討

- 176-205行目の最外ループを並列化すると配列Pが正しく計算されない(配列Pはループ変数`loop`の次元を持たないので, 並列化すると配列Pに逐次計算と同じ結果が格納されない).
- 177-195行目と197-203行目のループを並列化の対象とする.
- ループ変数`k`の次元を有する配列がマッピングの対象となる.

配列の宣言内容	配列名
(mimax,mjmax,mkmax)	p,bnd,wrk1,wrk2
(mimax,mjmax,mkmax,4)	a
(mimax,mjmax,mkmax,3)	b,c

- DISTRIBUTE指示文を挿入する(メインプログラムとサブルーチンinitに対して同様に挿入する). エディタを用いてhimeno.fを修正する.
% vi himeno.f (エディタはviの他にも利用可能)

4. HPFプログラミング演習(1) Step-2

- DISTRIBUTE指示文の挿入.

```
common /pres/ p(mimax,mjmax,mkmax)
      common /mtrx/ a(mimax,mjmax,mkmax,4),
+      b(mimax,mjmax,mkmax,3),c(mimax,mjmax,mkmax,3)
      common /bound/ bnd(mimax,mjmax,mkmax)
      common /work/ wrk1(mimax,mjmax,mkmax),wrk2(mimax,mjmax,mkmax)
CC other constants
      common /others/ imax,jmax,kmax,omega
!HPF$ DISTRIBUTE (*,*,BLOCK):: p,bnd,wrk1,wrk2
!HPF$ DISTRIBUTE (*,*,BLOCK,*) :: a,b,c
```

- 177-195行目のループには変数gosaの総和を計算する処理があるので、INDEPENDENT指示文でREDUCTION指定を行う.

```
!HPF$ INDEPENDENT,REDUCTION(gosa)
      DO K=2,kmax-1
        DO J=2,jmax-1
          DO I=2,imax-1
            S0=a(I,J,K,1)*p(I+1,J,K)+a(I,J,K,2)*p(I,J+1,K)
            (中略)
            GOSA=GOSA+SS*SS ← 総和計算
            wrk2(I,J,K)=p(I,J,K)+OMEGA *SS
          enddo
        enddo
      enddo
```

4. HPFプログラミング演習(1) Step-3

- 合計7行の指示文を追加したら、コンパイルを実行する。-Mlist2オプションを追加して並列化やデータ転送の有無を確認する。

% sxhpf -cfsx9 -Mlist2 himeno.f

- (参考)-Mlist2オプションによる並列化情報リストfilename.lstの見方

並列化できないと判定されたループ(<S>)

並列化可能なら、INDEPENDENT指示文を指定すれば並列化される可能性あり

```
( 1 )      subroutine sub(a,inew,iold)
( 2 )      real a(100,100,2)
( 3 )      !HPF$ DISTRIBUTE a(*,BLOCK,*)
( 4 ) <S>----- do j=1,100
      COMM: SFT [a] [LINO: 5 in sample.F]
( 5 ) <N>----- do i=1,100
( 6 ) |          a(i,j,inew)=a(i,j-1,iold)+a(i,j,iold)
( 7 ) +----- enddo
( 8 )      enddo
( 9 )      end
```

通信発生箇所(COMM:)

性能低下の原因箇所

並列化可能だが並列化されなかったループ(<N>)

データの分散を変えれば、並列化される可能性あり

4. HPFプログラミング演習(1) Step-3

himeno.lstを確認

```
( 182) <S>----- DO loop=1,nn
( 183) !HPF$ INDEPENDENT,REDUCTION(gosa)
      COMM: RED [gosa] [LINO: 184 in himeno-2.f]
      COMM: SFT [p] [LINO: 184 in himeno-2.f]
      HOME: a(:, :, k, :)
( 184) <P>----- DO K=2, kmax-1
( 185)   <I>----- DO J=2, jmax-1
( 186)   <I>----- DO I=2, imax-1
( 187)             S0=a(I, J, K, 1)*p(I+1, J, K)+a(I, J, K, 2)*p(I, J+1, K)
( 188)             1      +a(I, J, K, 3)*p(I, J, K+1)
( 189)             2      +b(I, J, K, 1)*(p(I+1, J+1, K)-p(I+1, J-1, K))
( 190)             3      -p(I-1, J+1, K)+p(I-1, J-1, K))
( 191)             4      +b(I, J, K, 2)*(p(I, J+1, K+1)-p(I, J-1, K+1))
( 192)             5      -p(I, J+1, K-1)+p(I, J-1, K-1))
( 193)             6      +b(I, J, K, 3)*(p(I+1, J, K+1)-p(I-1, J, K+1))
( 194)             7      -p(I+1, J, K-1)+p(I-1, J, K-1))
( 195)             8      +c(I, J, K, 1)*p(I-1, J, K)+c(I, J, K, 2)*p(I, J-1, K)
( 196)             9      +c(I, J, K, 3)*p(I, J, K-1)+wrk1(I, J, K)
( 197)             SS=(S0*a(I, J, K, 4)-p(I, J, K))*bnd(I, J, K)
( 198)             GOSA=GOSA+SS*SS
( 199)             wrk2(I, J, K)=p(I, J, K)+OMEGA *SS
( 200)           enddo
( 201)         enddo
( 202)       +----- enddo
      HOME: p(:, :, k)
( 204) <P>----- DO K=2, kmax-1
( 205)   <I>----- DO J=2, jmax-1
( 206)   <I>----- DO I=2, imax-1
( 207)             p(I, J, K)=wrk2(I, J, K)
( 208)           enddo
( 209)         enddo
( 210)       +----- enddo
( 212)     enddo
```

4. HPFプログラミング演習(1) Step-4

■ HPFプログラムの実行

- 実行用スクリプトをrun2.shとして用意している(4プロセスでの実行).

% qsub run2.sh

- 実行結果を確認する. 実行結果はstep-4.oXXXX(XXXXにはジョブIDが入ります)として格納される.

% cat step-4.oXXXX

```
mimax= 513   mjmax= 257   mkmax= 257
imax= 512   jmax= 256   kmax= 256
Start rehearsal measurement process.
Measure the performance in 1000 times.
MFLOPS: 96206.80   time(s): 11.62815403938293  0.6827177
```

➤ 43.2秒が11.6秒に短縮.

5. 並列化情報リストの利用

■ 並列化情報リストから問題を抽出する(1)

◆ 並列化できないと判定されたループの抽出方法

```
%>grep "<S>" filename.lst  
( 57) <S>----- do ij=2,ibar*2,2
```

57行目のdo ijのループは並列化できないと判定された

- 特に、コストの高い通信を伴う場合、注意が必要。INDEPENDENT指示文(+NEW節、REDUCTION節)を指定すると並列化される場合が多い
- 同じループネストに **<P>** (並列化された)マークのついたループがあり、通信がない場合は問題ない

```
(131) <P>----- do k=1,ibar  
(132) | <S> ----- do n=1,n
```

◆ 並列化可能だが並列化されなかったループの抽出方法

- ループ中の配列を分散すると、並列化される場合あり
- 通信を伴わない場合は、問題ないことが多い

```
%>grep "<N>" filename.lst  
( 44) <N>----- do i=0,ib
```

44行目のdo iのループは並列化できるが並列化しなかった

5. 並列化情報リストの利用

■ 並列化情報リストから問題を抽出する(2)

◆ 通信発生箇所の抽出方法

```
%> grep COMM: finename.lst  
COMM: SCL [u] [LINO: 137 in filename.hpf]
```

◆ 出力フォーマット

COMM: **通信パタン** [変数名] [LINO: 行番号 in ファイル名]

◆ 通信パタンとしては、以下のものがある

- ・ マッピング(DISTRIBUTE指示文等)の追加・修正やINDEPENDENT指示文による並列化、ON指示構文+LOCAL節による通信抑制などで削減できる場合も多い

SFT シフト通信。コストが低いので通常は問題ない。

RED 集計計算。並列ループの外側に生成されているなら通常やむを得ない。
並列ループの内側ならREDUCTION節付きINDEPENDENT指示文が有効。

SCL 一要素通信。配列が対象の場合は、通信コストが高いため**チェックが必要**。

CPY **配列のテンポラリへのコピー**。通信対象の配列と、ループ並列化の基準配列との間にマッピングの不整合がある場合等に発生する。

G/S Gather/Scatter。間接アクセスループ等で発生し**通信コストが高い**場合が多い

6. クリーンナップとデバッグ

■ 補足情報として, HPFプログラムのクリーンナップとデバッグについて紹介する.

マップできない配列

- 他の変数とメモリ領域を共有する配列はマップできない。(翻訳時にエラーとなる)

◆ EQUIVALENCE文に記述した配列

修正方法: 作業配列の利用・再利用や、動的に大きさを決めたい配列には、**割付け配列**や**自動割付け配列**を使う

ある型の配列を別の型で参照するようなプログラミングは避ける。

◆ POINTER/TARGET属性を持つ配列

修正方法: 動的にメモリを確保したい場合には、**割付け配列**や**自動割付け配列**を利用する。

● 自動割付け配列とは:

```
SUBROUTINE SUB(N)
```

```
COMMON /COM/ M
```

```
REAL A(N,M) ! 大きさが、引数や共通ブロック変数により実行時に決まる局所配列
```

Fortranとの非互換事項(1)

- 手順間で結合する配列の形状・型等は、原則として同一にする
- HPFの文法違反の例

◆ 引数結合(実引数と仮引数)

例1: アドレス渡し

```
real a(100,100)
!hpf$ distribute a(*,block)
call sub(a(1,i))
end
subroutine sub(a)
real a(100)
```

例2: 実引数と仮引数の形状が異なる

```
real a(10000)
call sub(a, 10)
end
subroutine sub(a,n)
real a(n,n)
!hpf$ distribute a(*, block)
```

◆ 共通ブロック

例1: マッピング指定漏れ

```
common /com/ a(100,100)
!hpf$ distribute a(*,block)
end
subroutine sub
common /com/a(100,100)
```

例2: 使わない手順で宣言省略

```
common /com/ a(100),b(100)
!hpf$ distribute (block) :: a,b
end
subroutine sub
common /com/a(100)
!hpf$ distribute (block) :: a
```

引数結合(1) HPF/SX V2のデバッグ機能

- 実行時に、エラーが発生した手続や対象引数の情報が標準エラー出力に出力される

例1: アドレス渡し

```
real a(100,100)
!hpf$ distribute a(*,block)
call sub(a(1,i))
end
subroutine sub(a)
real a(100)
```

例2: 実引数と仮引数の形状が異なる

```
real a(10000)
call sub(a,10)
end
subroutine sub(a,n)
real a(n,n)
!hpf$ distribute a(*,block)
```

例1: "仮引数名": Nonsequential dummy **array** is associated with **array element or scalar actual**. PROG="手続名" ELN="行番号"


例2: "仮引数名": Dummy argument **rank differs** from actual. PROG="手続名" ELN="行番号"

引数結合(2) 修正方法1

対応する実引数と仮引数の形状や型等是一致的させる

例1:


```
real a(100,100)
!hpf$ distribute a(*,block)
do l=1,100
  call sub(a(1,l)) ! アドレス渡し
enddo
end
subroutine sub(a)
real a(100)
```



```
real a(100,100)
!hpf$ distribute a(*,block)
do l=1,100
  call sub(a(:,l)) ! 部分配列
enddo
end
subroutine sub(a)
real a(100)
```

例2:

```
real a(100000) ! 大きめに
call sub(a,10)
end
subroutine sub(a,n)
real a(n,n)
```



```
real, allocatable :: a(:,:) ! 割付け配列
read(5) n
allocate(a(n,n)) ! 使用する形状で割付け
call sub(a,n)
end
subroutine sub(a,n)
real a(n,n)
```

引数結合(3) 修正方法2

- 大きさ引継ぎ配列(擬寸法配列)は、最後の次元の寸法がないためマップしたり、マップされた実引数と結合できない

→ **整合配列、又は形状引継ぎ配列**を利用する

例:


```
real a(100,100)
call sub(a)
end
```

```
subroutine sub(a)
real a(100,*)
```

整合配列

```
real a(100,100)
call sub(a,100)
end
```

```
subroutine sub(a,n)
real a(100,n)
```



形状引継ぎ配列:

Interfaceブロック等により、呼出し側で、呼ばれ側手続の引数が形状引継ぎ配列であることを明示する必要がある(Fortranの仕様)

```
real a(100,100)
interface
  subroutine sub(a)
    real a(:, :)
  end subroutine
end interface
call sub(a)
end

subroutine sub(a)
real a(:, :)
```


共通ブロック(1) HPF/SX V2のデバッグ機能

翻訳時オプション-Mcommonchkを指定して翻訳すると、
実行時に不整合が検出され、
修正すべき手続名や配列名が標準エラー出力に表示される

注意 本オプションを指定する場合、全ての手続に対して指定する

例1: マッピング指定漏れ

```
common /com/ a(100,100)
!hpf$ distribute a(*,block)
end
subroutine sub
common /com/a(100,100)
```

例2:使わない手続で宣言省略

```
common /com/a(100),b(100)
!hpf$ distribute (block) :: a,b
end
subroutine sub
common/com/ a(100)
!hpf$ distribute (block) ::a
```

例1: Inconsistency detected in **the mapping attribute** of common variable
between “手続名” and “手続名” : “配列名” in /共通ブロック名/

例2: Inconsistency detected in **the number of explicitly mapped arrays** of common
block between “手続名” and “手続名” : “配列名” in /共通ブロック名/

共通ブロック(2) 修正方法

共通ブロックの宣言は、全手続で一致させる(マッピングを含む)

大域変数は**モジュール**やINCLUDEファイル中に宣言すると、
宣言の記述が一回で済むため、書き忘れや書き誤りが防げる

例 : **モジュールによる大域変数の宣言** **INCLUDEファイルによるCOMMONの宣言**

```
module com1
dimension a(100,100)
!hpf$ distribute (*,block) :: a
end module

module com2
dimension b(100,100),c(100,100)
!hpf$ distribute (*,block) :: b,c
end module
```

```
subroutine sub()
use com1
use com2
end

subroutine sub2()
use com1
use com2
end
```

```
%> cat com1.h
common /com1/ a(100,100)
!hpf$ distribute (*,block) :: a

%> cat com2.h
common /com2/ b(100,100),c(100,100)
!hpf$ distribute (*,block) :: b,c
```

```
subroutine sub()
include "com1.h"
include "com2.h"
end

subroutine sub2()
include "com1.h"
include "com2.h"
end
```

Fortranとの非互換事項(2)

■ 宣言範囲外アクセス

- ◆ プログラムミスその他、ベクトル長を稼ぐためにループを一重化する場合などにも発生する

例: ループ一重化

```
real a(100,100)
!hpf$ distribute a(*,block)
do i=1,10000
  a(i,1) = ...
enddo
```

・分散次元より前の場合は可能

```
real a(100,100,100)
!hpf$ distribute a(*,*,block)
do k=1,100
  do i=1,10000
    a(i,1,k) = ...
  enddo
enddo
```

- ◆ システム領域が破壊され、全く別の場所や、ずっと後のステップでアボートする場合がある。

宣言範囲外アクセス：HPF/SX V2のデバッグ機能

■ 宣言範囲外アクセス検出機能

翻訳時オプション-Msubchkを指定して翻訳すると、
実行時に宣言範囲外アクセスの有無がチェックされる

並列化・ベクトル化は可能な限り行われる

範囲外アクセス検出時、配列名、行番号等の情報が警告又は致命的
エラーメッセージとして標準エラー出力に出力される(既定値は警告)。

・実行時オプション-hpf -subchk fatal を指定すると、
範囲外アクセス検出時に実行を終了させることもできる。

例：実行時オプションの指定方法： `mpirun -nn 2 -nnp 8 ./a.out -hpf -subchk fatal`

```
real a(100,100)
!hpf$ distribute a(*,block)
do i=1,10000
  a(i,1) = ...
enddo
```

“配列名” is accessed **out of declared bounds** along %d th dim
PROG=“手続名” ELN=“行番号”

7. 実行時性能情報の取得方法

■ 主に3つの手段がある

MPIPROGINF: プログラム全体の情報

- ◆特別な翻訳時オプションは不要。**測定オーバーヘッドなし。**
- ◆使い方はMPIの場合と同じ。
- ◆実行時環境変数MPIPROGINFを指定して実行する

ftrace: 手続毎の情報

- ◆翻訳時オプション-ftraceを指定して実行する
- ◆使い方はMPIの場合と同じ

```
%>sxftrace -f ftrace.out.*
```

etime: 計時用手続。任意の区間の実行時間(経過時間)

- ◆使い方はFortranの場合と同じ
- ◆並列ループ中から呼び出すことは出来ない

8. HPFプログラミング演習(2)

■ 3つのプログラム単位: 全86行(空白行とコメントを含む)

- モジュールparam

- 問題サイズと時間発展ループの繰返し数を大域定数として宣言

- 主プログラムsample

- 主要配列は2次元: $a(n,n)$, $b(n,n)$, $c(n,n)$

- 初期化後、時間発展ループ中で計算を行い、実行時間とチェックsumを出力

- サブルーチンbound

- 境界処理ループだけを含む

備考: 頻出パタンの練習です。計算内容に意味はありません。

■ 既にコードのクリーンアップ後の状態

- このままでもHPFコンパイラで翻訳、実行は可能 (もちろん、何台で実行しても速くはない)

8. HPFプログラミング演習(2)

■ プログラムをコンパイルして実行する

- ディレクトリを移動する.

% cd hpf/practice_2

- FTRACE情報を採取して演算コストを把握する.

% sxf90 -ftrace -cfsx9 sample.F

- 実行スクリプトを用意していますので, サブミットしてSX-9で実行する.

% qsub run.sh

- 実行結果を確認する. 実行結果はpractice_2.oXXXX(XXXXはジョブID)として格納される.

% cat practice_2.oXXXX

8. HPFプログラミング演習(2)

- もっともコストの高いループは以下.

```
! main loop
  do j=2, n-1
    do i=2, n-1
      ix = idxx(i)
      iy = idxy(j)
      a(i,j) = (b(i,j)+b(i-1,j)+b(i+1,j)+b(i,j-1)+b(i,j+1))*0.2d0*c(ix,iy)*ap
    enddo
  enddo
```

PROC.NAME	FREQUENCY	EXCLUSIVE TIME[sec](%)	AVER.TIME [msec]	MOPS	MFLOPS	V.OP RATIO	AVER. V.LEN	VECTOR TIME	I-CACHE MISS	O-CACHE MISS	BANK CPU	CONFLICT PORT	NETWORK
sample	1	6.120(100.0)	6120.403	18109.3	6174.5	99.72	253.3	6.114	0.000	0.000	0.364	4.577	
bound	1002	0.001(0.0)	0.001	5602.9	0.0	98.61	255.9	0.001	0.000	0.000	0.000	0.001	

total	1003	6.122(100.0)	6.104	18106.3	6173.0	99.72	253.3	6.115	0.000	0.000	0.364	4.578	

8. HPFプログラミング演習(2)

■ 並列化するループ(35行目)中で定義される配列をマップする

- do **j**のループがアクセスする、配列**a**の2次元目をマップする
 - 通常は、BLOCK分散が効率が良い
 - 問題サイズの関係上、BLOCK分散では負荷バランスが不均等になる場合等、GEN_BLOCK分散が有効な場合もある。
 - 本ループだけを見れば、右辺の配列**b,c**はマップする必要はない。但し、マップしても通信が出ない場合は、マップしたほうが**メモリの節約**になる

```
!hpf$ distribute (*,block) :: a
:
! main loop
  do j=2, n-1
    do i=2, n-1
      ix = idxx(i)
      iy = idxy(j)
      a(i,j) = (b(i,j)+b(i-1,j)+b(i+1,j)+b(i,j-1)+b(i,j+1))*0.2d0*c(ix,iy)*ap
    enddo
  enddo
```

8. HPFプログラミング演習(2) アクセスパターン表

- 各配列に対して、並列化対象ループがアクセスする次元の、添字の一覧表を作成すると、最適なデータマッピングの候補を決めるのに便利
- 例:

```
do j=2, n-1 ! 並列化対象
  do i=2, n-1
    ix = idxx(i)
    iy = idxy(j)
    a(i,j) = (b(i,j)+b(i-1,j)+b(i+1,j)+b(i,j-1)+b(i,j+1))*0.2d0*c(ix,iy)*ap
  enddo
enddo
```

参照種別 \ 配列名	a	b	c	idxx	idxy
定義	j				
使用		j, j-1, j+1	*	*	j

- 並列化対象ループ(do j)がアクセスする次元の添字を、定義と使用に分けて一覧に。
- 配列cやidxxは、並列化対象DOループのDO変数jに対応する次元がないため、「*」と表示。

既にデータマッピングを決定した配列aを基準とすると

•配列bは、

- ALIGN b(j) WITH a(j)
- ALIGN b(j) WITH a(j-1)
- ALIGN b(j) WITH a(j+1)
- 非分散

のいずれかが最適

•配列idxyは、

- ALIGN idxy(j) WITH a(j)
- 非分散

のいずれかが最適

•配列c、idxxは、

•非分散が最適

※実際のALIGNの指定には
テンプレートが必要

8. HPFプログラミング演習(2)

■ マップした配列aが出現するループを探す(50行目)

- 左辺に出現する配列bを配列aに合わせてマップする
 - 分散次元の宣言範囲が同一で、アクセスにもずれがない(共にj)ため、`distribute`指示文だけでよい
 - もちろん、`align`指示文や`template`指示文を利用して、同等のマッピングを指定することも可能


```
double precision a(n,n),b(n,n)
!hpf$ distribute (*,block) :: a,b
:
do j=2, n-1
  do i=2, n-1
    ix = idxx(i)
    b(ix,j) = a(i,j) * c(i,j)
    ap = ap * a(i,j)
  enddo
enddo
```

- ここで、配列bのデータマッピングは、前頁のアクセスパタン表から導出される候補にも適合していることに注意。

8. HPFプログラミング演習(2)

■ 仮引数をマップした手続boundの呼出箇所をチェックする

◆ 実引数が配列aの場合(48行目)は、仮引数とマッピングが同一



◆ 実引数が配列cの場合(28行目)は、配列cがマップされていないので、サブルーチンboundの呼出し時と戻り時に通信が発生

- 実配列引数cを仮引数に合わせてマップするか、サブルーチンboundのクローニングを行うかを決める。
- この作業を忘れた場合、実行時オプション-hpf -commmsgを付けて実行すると、手続呼出し時に通信/コピーが発生する際、メッセージが出力されるのでチェック可能。

8. HPFプログラミング演習(2)

- 配列cはメインループ中で間接アクセスされているので、サブルーチンboundの仮引数に合わせてマップすると、配列aやbとアクセスパターンが一致せず通信が発生
 - 配列cはマップしない (非分散とする。p.5のアクセスパターン表から導出されるマッピングの候補にも適合していることに注意)
 - サブルーチンboundのクローニングを行う

```
double precision a(n,n),b(n,n),c(n,n)
!hpf$ distribute (*,block) :: a,b
.
! main loop
  do j=2, n-1
    do i=2, n-1
      ix = idxx(i)
      iy = idxy(j)
      a(i,j) = (b(i,j)+b(i-1,j)+b(i+1,j)+b(i,j-1)+b(i,j+1))*0.2d0*c(ix,iy)*ap
    enddo
  enddo
```

8. HPFプログラミング演習(2)

- 仮引数をマップしていないこと以外、サブルーチンboundと同一のサブルーチンbound_nodistを作成し、boundの実引数が配列cの場合(28行目)、bound_nodistの呼出しに置き換える

```
double precision a(n,n),b(n,n),c(n,n)
!hpf$ distribute (*,block) :: a,b
:
call bound(b)
call bound_nodist(c)
:
call bound(a)
:
end
subroutine bound(a)
double precision a(n,n)
!hpf$ distribute (*,block) :: a
:
end
subroutine bound_nodist(a)
double precision a(n,n)
:
end
```

8. HPFプログラミング演習(2)

- -Mftn、-Mlist2、-Minfoオプション付きで翻訳する
 - ◆ 並列化情報リストをチェックする (-Mlist2オプションで生成される)
 - -Minfo:診断メッセージを出力、-Mftn:F90/SXによる翻訳・リンクを省略

```
%> sxhpf -Mftn -Mlist2 -Minfo sample.F
:
%> grep "<S>" sample.lst
( 35) <S>----- do iter=1,maxiter
( 50) <S>----- do j=1,n
( 51) <S>----- do i=1,n
%> grep COMM: sample.lst
COMM: SFT [b] [LINO: 38 in sample.F]
COMM: CPY [a] [LINO: 54 in sample.F]
COMM: SCL [a] [LINO: 54 in sample.F]
COMM: RED [bp] [LINO: 55 in sample.F]
COMM: RED [sum] [LINO: 64 in sample.F]
COMM: SFT [a] [LINO: 83 in sample.F]
COMM: SFT [a] [LINO: 83 in sample.F]
```

並列化できないと判定されたループの抽出

時間発展ループ。並列化できない。

50,51行目のループネストは要チェック

通信発生箇所の抽出

- SFT(シフト)は多くの場合問題なし
- RED(集計)は多くの場合やむを得ない

要チェック。上記の50、51行目の<S>に対応して出ている可能性あり

- CPY(コピー)は、テンポラリへの置き換え(通常通信を伴う)
- SCL(要素毎)は、配列が対象の場合、高コスト


8. HPFプログラミング演習(2)

■ 配列a,bの分散次元に対応するdo jで並列化可能だが、コンパイラは、自動判定できていない

- do jのループにindependent指示文を指定する
- 但し、変数apに対する集計計算(総積)を含むため、reduction節も必要
 - reduction節を忘れると結果不正となる

・並列化情報リスト

```
!hpf$ distribute (*,block) :: a,b
( 50) <S>----- do j=1,n
      COMM: CPY [a] [LINO: 54 in sample.F]
      COMM: SCL [a] [LINO: 54 in sample.F]
( 51) <S>----- do i=1,n
      COMM: RED [ap] [LINO: 55 in sample.F]
( 52)                ix = idxx(i)
( 53)                b(ix,j) = a(i,j)*c(i,j)
( 54)                ap = ap * a(i,j)
( 55)                enddo
( 56)                enddo
```



```
!hpf$ distribute (*,block) :: a,b
:
!hpf$ independent, reduction(ap)
do j=1,n
  do i=1,n
    ix = idxx(i)
    b(ix,j) = a(i,j)*c(i,j)
    ap = ap * a(i,j)
  enddo
enddo
```


8. HPFプログラミング演習(2)

■ 再度、-Mftn、-Mlist2、-Minfoオプション付きで翻訳する

◆ 並列化情報リストをチェックする

```
%> sxhpf -Mftn -Mlist2 -Minfo sample.F
:
%> grep "<S>" sample.lst
( 35) <S>----- do iter=1,maxiter
( 55) <S>----- do i=1,n
%> grep "<P>" sample.lst
( 38) <P>----- do j=2,n-1
( 47) <P>----- do i=1,n
( 54) <P>----- do j=1,n
( 65) <P>----- do j=1,n
%> grep COMM: sample.lst
COMM: SFT [b] [LINO: 38 in sample.F]
COMM: RED [bp] [LINO: 54 in sample.F]
COMM: RED [sum] [LINO: 65 in sample.F]
COMM: SFT [a] [LINO: 84 in sample.F]
COMM: SFT [a] [LINO: 84 in sample.F]
```

並列化できないと判定されたループの抽出

時間発展ループ。並列化できない。

同じループネストの54行目の<S>が消え、<P>に変わっており、重い通信も出ていないので問題なし

並列化されたループの抽出

通信発生箇所の抽出

•CPY/SCLマークは消滅

8. HPFプログラミング演習(2)

■ 76行目のループは、シフト通信(SFT)が発生している境界処理ループ


- ◆ 配列aの分散次元の寸法は $n=2047$ なので、例えば16プロセッサで並列実行する場合、 $a(:,1)$ と $a(:,2)$ 、および $a(:,n)$ と $a(:,n-1)$ は同一のプロセッサ上にある

➢ ON HOME指示構文+LOCAL節を指定すれば通信の抑制が可能

● 境界処理で頻出のパターン

・並列化情報リスト

```
double precision a(n,n)
!hpf$ distribute (*,block) :: a
.
COMM: SFT [a] [LINO: 84 in sample.F]
COMM: SFT [a] [LINO: 84 in sample.F]
( 76) <N>----- do i=1,n
( 77) |             a(i,1) = a(i,2)
( 78) |             a(i,n) = a(i,n-1)
( 79) +----- enddo
```



```
double precision a(n,n)
!hpf$ distribute (*,block) :: a
do i=1,n
!hpf$  on home(a(:,1)), local begin
a(i,1) = a(i,2)
!hpf$  endon
!hpf$  on home(a(:,n)), local begin
a(i,n) = a(i,n-1)
!hpf$  endon
enddo
```

8. HPFプログラミング演習(2)

■ 再度、-Mftn、-Mlist2、-Minfoオプション付きで翻訳する

◆ 並列化情報リストをチェックする

```
%> sxhpf -Mftn -Mlist2 -Minfo sample.F
```

```
:
```

```
%> grep "<S>" sample.lst
```

```
( 35) <S>----- do iter=1,maxiter
```

```
( 55) <S>----- do i=1,n
```

```
%> grep "<P>" sample.lst
```

```
( 38) <P>----- do j=2,n-1
```

```
( 47) <P>----- do i=1,n
```

```
( 51) <P>----- do j=1,n
```

```
( 65) <P>----- do j=1,n
```

```
%> grep COMM: sample.lst
```

```
COMM: SFT [b] [LINO: 38 in sample.F]
```

```
COMM: RED [bp] [LINO: 51 in sample.F]
```

```
COMM: RED [sum] [LINO: 65 in sample.F]
```

並列化できないと判定されたループの抽出

時間発展ループ。並列化できない。

同じループネストの54行目の<S>が消えて、並列化(<P>)されており、重い通信も出ていないので問題なし

並列化されたループの抽出

通信発生箇所の抽出

・CPY/SCLマークは消滅

76行目(境界処理部分)のSFT(シフト通信)が消滅

8. HPFプログラミング演習(2)

■ HPFプログラムの実行

- 実行用スクリプトをrun2.shとして用意している(4プロセスでの実行).

% qsub run2.sh

- 実行結果を確認する. 実行結果はpractice_2.oXXXX(XXXXXはジョブID)として格納される.

% cat practice_2.oXXXX

9. HPFプログラミングのテキスト紹介

http://www.hpfpc.org/book/hpf_text.html



培風館 2011年3月30日発売
ISBN978-4-563-01586-2 ¥3400+税

HPFによる並列プログラミング手法唯一の実践的な解説書。(文法書に類するものはあった)

High Performance Fortran 2.0公式マニュアル
(シュプリンガー・フェアラク東京)
The High Performance Fortran Handbook
(The MIT Press)

**対象
読者**

- ・(分散)並列計算に興味のある初心者から上級者まで
- ・初歩的なFortranの知識は仮定(DO文, 代入文, IF文程度)

著者


監修 津田孝夫
編集 HPF推進協議会
著者 岩下英俊・坂上仁志・妹尾義樹・林康晴

さいごに

「MPIの 20%の手間 で 80%の性能」

プログラム開発・保守を効率化

本来の研究に注力



Empowered by Innovation **NEC**



Empowered by Innovation

NEC